
[bookmark: _Toc34646200]INTRODUCTION
With ICD you can debug COBOL routines which have been bound to a COBOL or ALGOL host program. Normally you will debug a single COBOL subprogram bound to a host. But you are not limited to this. You may debug all your COBOL subprograms if you like. You can even debug the COBOL host at the same time.
We will begin by discussing how to use ICD with COBOL bound routines. Later we will discuss an example of debugging routines bound to an ALGOL host program.
The following notes assume that you are already familiar with ICD and how to compile using the ICD compiler. If you have not used ICD before, please do so before attempting to use it on bound routines. This discussion also assumes that you are familiar with BINDING (See the A Series Binder Programming Reference Manual). You should be familiar with the terms used here from reading the ICD and BINDER manuals.
[bookmark: _Toc34646201]Restrictions on ICD commands

In implementing debugging some limitations on the normal functionality of ICD were necessary. Many of these restrictions apply only when you are trying to debug more than one COBOL routine (as used in this document routine refers to a COBOL subprogram or the COBOL host). If you are debugging just a single COBOL routine then you can ignore many of the restrictions discussed here.
[bookmark: _Toc34646202]Display of global variables
ICD can display all the variables syntactically visible to your COBOL subroutine -- that is it can display variables declared local to the subroutine and any global variables visible in the host. However if you need a USE statement in your bind deck to resolve naming differences between host and subprogram then you will not be able to view those global variables with ICD.
[bookmark: _Toc34646203]Breakpoints
You can set breakpoints in any COBOL routine that has been compiled with $ICD set. However, if you are setting breakpoints in a different routine these breakpoints cannot be verified at the time you set them. ICD assumes that breakpoints you set reference the routine you are currently in. The syntax for setting breakpoints in a different routine is explained below. 
If you have a source line number that appears in more than one routine, there is no way to specify that this breakpoint applies just to a specific bound routine. See the breakpoint discussion below for further details.
[bookmark: _Toc34646204]WHEN statement
The global WHEN statement evaluates an expression at the beginning of each COBOL statement. It might address local variables in the current routine. If you are debugging multiple routines bound to a host the global WHEN may have to be evaluated in a routine other than the one where the local variables were declared. ICD does not handle this situation and it will lead to unpredictable results. Use caution with global WHEN statements that reference local variables in any programs which have multiple bound routines.
You can still use WHENs attached to breakpoints if you are sure that the breakpoint line number is unique. This assures that the WHEN will only be evaluated at one line number - in the routine where it was declared.

[bookmark: _Toc34646205]BREAKPOINTS in multiple routines
You may have ICD enabled in more than one subprogram. With normal, non-bound, ICD you are used to an automatic break when the first statement of your Procedure Division is entered. With bound code there is only one such automatic break for the whole bound code file. This break will occur in the host program if it is compiled with ICD (and $ RESET ICDACTIVE is not set) or in the first subprogram that you enter that was compiled with ICD.
 Once you enter this first routine you will get the normal ICD greeting message. Thereafter you will only stop when you encounter a breakpoint or finish a CONTINUE + <num>command. In particular you will not automatically stop when you enter each subsequent bound routine for the first time.  If you want to be sure that you stop in a particular routine set a breakpoint there, use the CONT + <num> command, or interrupt using ?HI.
If you want to be sure to stop the next time you enter the subprogram you are currently in use BREAK BEGIN to set a break point at the first source line in the subprogram. In addition you might want to break just before dropping back out of the subprogram. If the last executable statement is also the last line of your source then using BREAK END is convenient in this case.
[bookmark: _Toc34646206]Setting breakpoints in other routines 
By default any breakpoints that you set are for the routine you are currently in. These breakpoints are verified and if they are not valid you receive a suitable warning message. For example, if you are in the host routine and enter BREAK 350200, the sequence number is checked to see that it is a valid breakpoint in the host and then placed in the active breakpoint list. The same thing happens if you are in a bound routine -- the breakpoint is checked to see if it is valid for that routine before it is placed in the active breakpoint list. 
ICD can only check the validity of sequence numbers for the routine where you are currently paused. Breakpoints in other routines cannot be verified. When it is necessary to set a breakpoint in another routine, ICD allows you to override the validity check by adding either IN MAINLINE or IN BOUNDROUTINE to the sequence number. The IN modifier means that you want to specify a breakpoint and bypass verification. 
Currently there is no difference in semantics between "IN BOUNDROUTINE" and "IN MAINLINE ". In each case the sequence number will simply be forced into the breakpoint list. For example, suppose you are just beginning your program and are paused in the mainline. If you want to set a breakpoint at a line number in a bound routine, enter: 
BREAK 320200 IN BOUNDROUTINE 
ICD responds with: 
BREAK inserted without verification
When your program encounters this line number it will pause. Please note that you cannot indicate a particular bound routine where this breakpoint applies. Because a breakpoint is based solely on a sequence number it applies to any routine, bound or main, which has this sequence number.
[bookmark: _Toc34646207]Distinguishing between breakpoints in host and bound routines 
Currently ICD does not distinguish between sequence numbers in the host and various bound routines when examining numbers in the breakpoint list. If a breakpoint is set at line 352300, for example, your program will stop when it encounters this line number -- whether in the host routine or any bound routine.
[bookmark: _Toc34646208]Stopping at the first or last line of a subprogram
BREAK BEGIN and BREAK END are convenient for setting breakpoints at the first and last valid source line numbers in your Procedure Division.
[bookmark: _Toc34646209]SOURCE FILE details
ICD may have up to 3 source files open at any one time. They are used to display your program source when you are at a breakpoint. These files are : the mainline or host file, the COPYLIB file, and the bound routine file. ICD tries to find the mainline source file automatically when your program starts. Similarly, as each bound routine is entered, ICD tries to find the source file used to compile it.
You may query the ICD source file settings at any time by simply entering the SOURCE command. The SOURCE command can also be used to change the file title that ICD uses to access any one of these three files.
[bookmark: _Toc34646210]Changing the Mainline source file title
You must be paused at a breakpoint in the mainline program to change the host program file title. While paused just enter SOURCE <file title>, for example:
SOURCE SYMBOL/COBOL/ACCOUNTING/UPDATE
Another alternative is to enter just the SOURCE command. ICD will display your current settings and ask you if you want to change anything. Entering YES at this point starts a dialogue which allows you to change any file titles.
[bookmark: _Toc34646211]Changing a COPYLIB file title
You change COPYLIB titles by entering dialogue mode in ICD. Enter just the SOURCE command. ICD will display your current settings and ask you if you want to change anything. Entering YES at this point starts a dialogue which allows you to change any file titles.
[bookmark: _Toc34646212]Changing the bound source file title
You must be paused at a breakpoint in a bound routine to change the bound program file title. While paused just enter SOURCE <file title>, for example:
SOURCE SYMBOL/COBOL/BINDABLE/UPDATEPROC

[bookmark: _Toc34646213]LIST command changes
In order to be able to list from the current bound routine source while you are in the host an additional modifier IN BOUNDROUTINE is allowed with the LIST command. For example:
LIST 0 FOR 10 IN BOUNDROUTINE
ICD will print 10 lines from the current bound routine source file. Similarly to list from the Host while in a bound routine just enter:
LIST 0 FOR 10 IN MAINLINE
Since ICD only handles the source from one bound routine at a time there is no ambiguity here. If you want to see another bound routine source file just use the SOURCE command (when you are paused in a bound routine) to change the setting for the bound file.
[bookmark: _Toc34646214]CONTINUE command changes
The continue command allows you to execute a number of statements and then regain control. You do this by entering something like: CONTINUE + 10, which will do 10 statements and then stop.
You can use this same form of the command when debugging bound routines with results identical to those you are used to with ICD. ICD counts statements as execution moves from host to bound routine and back. When the proper number of statements has been executed ICD regains control. For example, suppose you are stopped in a bound routine right at the point where it is to EXIT back to the mainline and you want to do a few more statements but you don't want to set a breakpoint. Just enter CONTINUE + 20 for example, and you will regain control after 20 statements -- either back in the mainline or in the next bound routine that is called from the mainline.
ICD only counts statements in routines where ICD and ICDACTIVE are enabled. If your host routine has ICDACTIVE reset or your host is an ALGOL program then statements will only be counted in those bound routines compiled with ICD.
[bookmark: _Toc34646215]EXAMPLES 
NOTE : The following example is a slight modification to the COBOL INTRALANGUAGE binding example found in A Series Binder Programming Reference Manual. In addition to the level 3 subprogram in that example, this host program calls an ALGOL subprogram (not shown) and another COBOL level 3 subprogram (not shown) which in turn calls a COBOL level 4 routine (not shown).
[bookmark: _Toc34646216]Remote file considerations
In bound code files all ICD dialog takes place through a remote file in the host. With COBOL hosts this file is either automatically declared for you (when you use the $ICD card) or you use an already present file. (See "Compiling Details" in the ICD manual for a detailed discussion.) For an ALGOL host the ICD dialog file is the one that you use in the ICD_ALGOL_INITIALIZE call. (See "ALGOL Host Example" in this manual for details.)
Normally any subprogram you compile with ICD will not have a remote input file. ICD will automatically use the host-declared file for dialog when in a subprogram. You do not to do anything special.
If any subprogram already uses a remote input file it will conflict with the ICD dialog file in the host. In this case your subprogram will have to declare its remote file as GLOBAL and link it to the remote file in the host with a Binder USE card. (See "ALGOL Host Example" in this manual for an example of this.)
[bookmark: _Toc34646217]COBOL Host Example
[bookmark: _Toc34646218]COBOL Host program
If you are going to debug any bound routines then you must, at a minimum, create the proper debugging environment in the host to allow the debugging of bound routines. This means that you need the $ICD card in the host and you must compile the host with the ICD compiler. However you do not have to actually debug the host. (Many on-line programs that need to be debugged use the host to dispatch to one of many subroutines based on the type of transaction entered. To debug these kinds of programs in an on-line environment you don't want debugging in the host. Rather you only want to debug one or more of the transaction subroutines).
There is a simple way to allow the host debug environment to be created while suppressing breakpoints in the host. Insert the $ICD card and in addition just insert a $ RESET ICDACTIVE card anywhere before the first line of Procedure Division code.
Use the $SET ICD card, including the name of any remote input file
 already declared in your host (see the ICD Manual for further details), and compile with the ICD compiler. If necessary, insert the RESET ICDACTIVE card anywhere before the first line of Procedure Division code. Then compile your host program as you would any other program that you are going to debug with ICD.
Suppose you have a mainline source program entitled SYMBOL/COBOL/HOST as follows:

200 $SET LISTDOLLAR
300 $SET ICD
305 *  % The reset ICDACTIVE is used to compile a host program for 
310 *  % BINDING that will establish the debug environment
315 *  % but not allow debugging in the host - just the bound routine
400 $ RESET ICDACTIVE
500 IDENTIFICATION DIVISION.
600 PROGRAM-ID. HOST.
700 ENVIRONMENT DIVISION.
800 CONFIGURATION SECTION.
900 SPECIAL-NAMES.
1000     "OBJECT"/"DEBUGGER"/"COBOL74"/"BINDING"/"BOUNDROUTINE"
1100          IS TO-BE-CALLED
1200     ,"OBJECT"/"DEBUGGER"/"COBOL74"/"BINDING"/"BOUNDROUTINELVL3"
1300           IS LEVEL3PROC-NAME
1305     ,"OBJECT/DEBUGGER/COBOL74/BINDING/ALGOLROUTINE"
1310           IS ALGOL-PROC-NAME.
1400 INPUT-OUTPUT SECTION.
1500 FILE-CONTROL.
1600     SELECT PR ASSIGN PRINTER.
1700 DATA DIVISION.
1800 FILE SECTION.
1900 FD PR.
2000 01  PR-RCD.
2100     03 FILLER PICTURE X(36).
2200 WORKING-STORAGE SECTION.
2300 77 GLOBAL-VAR1   COMP PICTURE 99.
2400 *
2500 *  When this array is declared in the bound routine it will have
2600 *  subordinate items declared to see if ICD can find them
2700 *
2800 01 GLOBAL-ARRAY.
2900 *    Make sure that the FILLER is big enough to hold the one
3000 *    declared in the bond routine. ICD uses the length of the
3100 *    bound declaration.
3200 03  FILLER PICTURE X(100).
3300 01 ORIG .
3400     03 FILLER PICTURE X(36).
3500 01 NEW  .
3600     03 FILLER PICTURE X(36).
3605 *   These two variables should be updated when the ALGOL routine
3610 *   is called
3615 77   MACHINETYPE   REAL.
3620 77   TOD           REAL.
3625 *
3630 *   These two variables are passed to the ALGOL routine and
3635 *   updated there and returned
3650 77   AA            REAL.
3655 77   BB            REAL.
3670 *
3700 LOCAL-STORAGE SECTION.
3800 LD PARMS.
3900 01 A REF.
4000    03 FILLER PICTURE X(36).
4100 01 B REF.
4200    03 FILLER PICTURE X(36).
4205 LD ALGOL-PARMS.
4210    77   AA           REAL REF.
4215    77   BB           REAL REF.
4300 PROCEDURE DIVISION.
4400  DECLARATIVES.
4500      S1 SECTION.
4600          USE EXTERNAL TO-BE-CALLED
4700             AS PROCEDURE WITH PARMS USING A B.
4800      LEVEL3-PROC SECTION.
4900          USE EXTERNAL LEVEL3PROC-NAME
5000             AS PROCEDURE WITH PARMS USING A B.
5005      ALGOL-PROC SECTION.
5010          USE EXTERNAL ALGOL-PROC-NAME
5015              AS PROCEDURE WITH ALGOL-PARMS USING AA BB.
5100  END DECLARATIVES.
5200  THE SECTION.
5300 START-IT.
6500      COMPUTE GLOBAL-VAR1 = 22.
6600      MOVE " Hello from the global array " TO GLOBAL-ARRAY.
6700 *
6800      OPEN OUTPUT PR.
7300 *
7400      MOVE "THIS WILL STOP WHEN THIS LINE ENDS" TO ORIG.
7405 *
7410 *     Call the ALGOL procedure. This procedure will 
7412 *     change its two parameters and also update
7413 *     the globals MACHINTYPE and TOD.
7415 *
7420      ENTER ALGOL-PROC USING AA, BB.
7425 *
7500      ENTER S1 USING ORIG NEW.
7600      WRITE PR-RCD FROM ORIG.
7700      WRITE PR-RCD FROM NEW.
7800 *
7900      MOVE 1 TO GLOBAL-VAR1.
8000 LOOP.
8100      ENTER S1 USING ORIG  NEW.
8200      ADD 1 TO GLOBAL-VAR1.
8300      IF GLOBAL-VAR1 LESS THAN 4 GO TO LOOP.
8400 *
8500 *
8600 *    Call a LEVEL 3 procedure which in turn calls a LEVEL 4 PROC
8700 *
8800      ENTER LEVEL3-PROC USING ORIG NEW.
8900 *
9000      STOP RUN.
Compile your program with the ICD compiler :
COMPILE SYMBOL/COBOL/HOST WITH ICDCOBOL74
[bookmark: _Toc34646219]COBOL Bound routine:
Include a $SET ICD card in your subprogram source and compile it with the ICD compiler. No changes need to be made to your routines. Of course a bound routine may not contain its own input remote file because this will interfere with the ICD dialogue file. Compile your routines as you normally would. For example suppose you have a source file titled SYMBOL/COBOL/BOUNDROUTINE as follows:

300 $SET OMIT
400 %%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%:
500 %-
600 %      There are 3 kinds of variables that can live in a COBOL
700 %    bound routine:
800 %          LOCAL - disappear on procedure exit
900 %          GLOBAL- Are variables which are declared in the host
1000 %                and which must be matched at bind-time.
1100 %          OWN   - Locally declared variables which are to be
1200 %                allocated at D2.
1399 %%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%:
1400 $POP OMIT
1500 $  SET ICD
1700 $SET LEVEL= 3
1800 IDENTIFICATION DIVISION.
1900 PROGRAM-ID. ARRAY-MIXER.
2000 ENVIRONMENT DIVISION.
2100 CONFIGURATION SECTION.
2200 DATA DIVISION.
2300 WORKING-STORAGE SECTION.
2400 *
2500 *     The following variable will be know only to this routine
2600 *         but will retain its value from call to call
2700 *
2800 77    VAR1  PICTURE  99 OWN.
2900 *
3000 *      The following variable is LOCAL.  LOCAL is redundant here
3100 *          unless the compiler option $GLOBAL is set.
3200 *
3300 77    VAR2  PICTURE  9(5) LOCAL.
3400 *
3500 *    The following variables are declared global.
3600 *    During binding they should be found in the host program.
3700 *     If the variable does not exist in the host then the BIND
3800 *     listing will have the message "<var> added to host"
3900 *
4000 $SET GLOBAL
4100 77 GLOBAL-VAR1 BINARY   PIC 9(11).
4125 01    GLOBAL-VAR3  PIC 999999.
4130 01    GLOBAL-VAR4  PIC X(10).
4200 $POP GLOBAL
4300 *
4400 *      More OWN variables
4500 *
4600 77    VAR3 PICTURE 999999 OWN.
			........
5500 77    VAR12 PIC 99999    OWN.
5600 *
5700 *
5800 *    END OF OWNS
5900 *
6000 01    GLOBAL-ARRAY        GLOBAL.
6100       03 FILLER        PICTURE X(5).
6105       03 XX OCCURS 5.
6110          05  GLOBAL-A OCCURS 3.
6115              07  GLOBAL-B PIC 99.
6120              07  GLOBAL-C PIC 99.
6125          05  GLOBAL-D PIC 9999.
6130       03 YY  PIC 99.
6200 *
6400 01 X REF.
6500      03 ONE  PICTURE  X(5).
6600      03 SECOND  PICTURE  X(5).
6700      03 THIRD  PICTURE  X(5).
6800      03 FOURTH  PICTURE  X(5).
6900      03 FIFTH  PICTURE  X(5).
7000      03 SIXTH  PICTURE  X(5).
7100      03 SEVENTH  PICTURE  X(5).
7200      03 EIGHTH  PICTURE  X(1).
7300 01 Y REF.
7400      03 FIRS  PICTURE  X(5).
7500      03 SECON  PICTURE  X(5).
7600      03 THIR  PICTURE  X(5).
7700      03 FOURT  PICTURE  X(5).
7800      03 FIFT  PICTURE  X(5).
7900      03 SIXT  PICTURE  X(5).
8000      03 SEVENT  PICTURE  X(5).
8100      03 EIGHT  PICTURE  X(1).
9000 PROCEDURE DIVISION USING X Y.
9100 THAT SECTION.
9200 MIX.
9300 *
9400 *    Touch all the variables so the compiler will allocate
9500 *    them at D2
9505     COMPUTE GLOBAL-VAR1 = 98765.
9510     COMPUTE GLOBAL-VAR3 = 12345.
9515     MOVE "Bill Graham   " TO GLOBAL-VAR4.
9700     MOVE 25 TO VAR1 , VAR2, .
9710     MOVE 25 TO  VAR12.
9800 *
9900 *  Put something in the global array
10000 *
10005   MOVE 25 TO GLOBAL-C (2,3).
10010   MOVE 2645 TO GLOBAL-D (4).
10015 *
10020 *
11800 THE-NEXT SECTION.
11900 P-1-NEXT-SECTION.
12000      MOVE ONE TO SECON.
12100      MOVE SECOND TO FOURT.
12200      MOVE THIRD TO FIRS.
12300      MOVE FOURTH TO THIR.
12400      MOVE FIFTH TO SIXT.
12500      MOVE SIXTH TO SEVENT.
12600      MOVE SEVENTH TO FIFT.
12700      MOVE EIGHTH TO EIGHT.
	Compile your program with the ICD compiler :
COMPILE SYMBOL/COBOL/BOUNDROUTINE WITH ICDCOBOL74
[bookmark: _Toc34646220]BINDING
Just bind your routines as you normally would. Suppose that the bind cards for this
 example are contained in a file called SYMBOL/COBOL/BIND CARDS :
100 HOST IS OBJECT/DEBUGGER/COBOL74/BINDING/HOST;
200 BIND S1 FROM OBJECT/DEBUGGER/COBOL74/BINDING/BOUNDROUTINE;
300 BIND LEVEL3-PROC 
350       FROM OBJECT/DEBUGGER/COBOL74/BINDING/BOUNDROUTINELVL3;
400 BIND LEVEL4-PROC 
425       FROM OBJECT/DEBUGGER/COBOL74/BINDING/BOUNDROUTINELVL4;
450 BIND ALGOL-PROC 
475       FROM OBJECT/DEBUGGER/COBOL74/BINDING/ALGOLROUTINE;
485 USE ALGOL-PROC FOR ALGOLPROCEDURE;
Begin binding with the following command:
	BIND SYMBOL/COBOL/BIND CARDS
Then run the program:
	RUN SYMBOL/COBOL/BINDCARDS

[bookmark: _Toc34646221]Binder Output
A partial bind output from this example follows:

BEGIN BINDING S1 OF CODE2150 FROM OBJECT/DEBUGGER/COBOL74/BINDING/BOUNDROUTINE,
      COBOL74 42.152, COMPILED 09/09/95 17:57:40
THE IDENTIFIER OF THE SEPARATE PROCEDURE DOES NOT MATCH THE DECLARATION IN THE HOST -- THIS IS A WARNING ONLY <<<<
STATEMENT REQUIRED: USE S1 FOR CODE2150;
GLOBAL-VAR1  <---- THIS NEW GLOBAL VARIABLE HAS BEEN ADDED TO THE HOST -- THIS IS A WARNING ONLY
      GLOBAL-VAR1  (02,0003)  HAS BEEN CHANGED TO  (02,0039)
GLOBAL-VAR3  <---- THIS NEW GLOBAL VARIABLE HAS BEEN ADDED TO THE HOST -- THIS IS A WARNING ONLY
      GLOBAL-VAR3  (02,0004)  HAS BEEN CHANGED TO  (02,003A)
GLOBAL-VAR4  <---- THIS NEW GLOBAL VARIABLE HAS BEEN ADDED TO THE HOST -- THIS IS A WARNING ONLY
      GLOBAL-VAR4  (02,0005)  HAS BEEN CHANGED TO  (02,003B)
      GLOBAL-ARRAY  (02,0007)  HAS BEEN CHANGED TO  (02,002A)
      ICD_LIBRARY  (02,0009)  HAS BEEN CHANGED TO  (02,0031)
      .ICD_CALL  (02,000A)  HAS BEEN CHANGED TO  (02,0033)
      .ICD_INITIALIZE  (02,000B)  HAS BEEN CHANGED TO  (02,0034)
. . .
[bookmark: _Toc34646222]ALGOL Host Example
[bookmark: _Toc34646223]ALGOL Host program:
You can debug COBOL routines which have been bound to an ALGOL host by adding a few declarations and a procedure call to your host source file. These things are normally done for you automatically when you compile a host program in COBOL with $SET ICD. In the case of ALGOL you must explicitly prepare the debugging environment. ALGOL hosts which call COBOL bound routines compiled with debugging enabled are required to do the following:
	declare a library (the ICD library)
	declare three entry points in the library
	declare a real variable immediately followed by an array 
	declare a remote input/output file if one does not already exist ( for ICD input/output ) 
	insert an initialize call to the ICD library before the first COBOL bound routine is called.
NOTE : This initialize call must be made from the outer block of the ALGOL host program and not from inside an initialize procedure as you normally would do. In other words the call must be made from level 2 code. You will get the message "UNEXPECTED STACK STRUCTURE" at run-time when you call the initialize procedure if you do not do this correctly.
The details of each requirement described below.
Declaring the ICD library
LIBRARY ICD_LIBRARY (TITLE = "OBJECT/DEBUGGER/LIBVSNnn.");

Replace nn by your current ICD version number.
NOTE : Make sure your ALGOL HOST declares the ICD library as ICD_LIBRARY. That is the name that the bound-in Cobol routines will be using. If you declare the library in the host as something else you will need a USE card in your bind deck like:
USE <ALGOL library internal name> AS ICD_LIBRARY;
Declaring three entry points in the library :
 
Note that there are two "initialize" procedures : ICD_ALGOL_INITIALIZE and ICD_INITIALIZE. The only procedure you will call from your ALGOL host is ICD_ALGOL_INITIALIZE .
PROCEDURE ICD_ALGOL_INITIALIZE (COUNT_WD, VERSIONWORD
                              , BREAKS, RMTF);
  VALUE VERSIONWORD;
  REAL COUNT_WD, VERSIONWORD; ARRAY BREAKS [0]; FILE RMTF;
         LIBRARY ICD_LIBRARY;
    
PROCEDURE ICD_INITIALIZE (COUNT_WD, VERSIONWORD    
                        , RMT_IN, RMT_OUT);
  VALUE COUNT_WD, VERSIONWORD;
  REAL COUNT_WD, VERSIONWORD; FILE RMT_OUT, RMT_IN;
  LIBRARY ICD_LIBRARY;
  
PROCEDURE ICD_CALL (SEQNUM);
  VALUE SEQNUM;
  REAL SEQNUM;
  LIBRARY ICD_LIBRARY;
Declaring a real variable immediately followed by an array

These declarations can appear anywhere in the outer block but it is critical that the two declarations appear in this order. The declarations are:
REAL COUNT_WORD;
ARRAY BREAK_ARAY [0:99];
You do not have to use these exact variables names. You will use these variables in the ALGOL host's initialize call to ICD.
Declaring a remote input/output file
This is the file where all the ICD dialog from the bound COBOL subroutines will take place. Although every bound routine compiled with $ICD gets an ICDREMTF file automatically declared that file is unused when bound. The ICD dialog takes place through the file declared in the ALGOL host and passed to ICD_ALGOL_INITIALIZE.
If the ALGOL host already has a remote I/O file then you can just use it. If you need to, declare the file as follows :
FILE REMOTE_FILE (KIND = REMOTE, MAXRECSIZE = 14, 
, MYUSE = IO);
Inserting an initialize call to the ICD library
Using the real variable, array and file variable you declared call the ALGOL initialize procedure:
ICD_ALGOL_INITIALIZE (COUNT_WORD, 0, BREAK_ARAY
, REMOTE_FILE);
The COUNT_WORD and BREAK_ARAY do not need to be initialized. Your ALGOL program may not alter them during execution because they are maintained by the ICD library and used by your COBOL bound routines compiled with ICD. When you run the bound version of your code file you will get the ICD greeting message when you enter the first COBOL bound routine that you compiled with $ICD set.
You can have multiple COBOL subroutines bound-in and compiled with debugging enabled. An example of a complete ALGOL host program follows:
[bookmark: _Toc34646224]ALGOL Host sample
This host may be called as a library, in which case its only function is to export the level 3 procedure. If the program is run normally then it calls the level 3 procedure immediately.

100 BEGIN
200 REAL VERSION_WORD;
300       
400     % ************ Beginning of ICD stuff ************
500    %%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%-
600    %-
700    %-                      WARNING
800    %-     These next two variables must be declared in this
900    %-     order with no intervening declarations
1000   %-
1100   %%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%-
1200 REAL
1300    COUNT_WORD;
1400 ARRAY
1500    BREAK_ARAY [0:99];
1600    %%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%-
1700    %-
1800    %-                 END WARNING
1900    %-     End of two variables which must be declared in this
2000    %-     order with no intervening declarations
2100    %-
2200    %%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%-
2300 FILE
2400    REMOTE_FILE (KIND=REMOTE, MAXRECSIZE=14, MYUSE=IO);
2500
2600 LIBRARY ICD_LIBRARY (TITLE = "OBJECT/DEBUGGER/LIBVSN16.");
2605
2700 PROCEDURE ICD_ALGOL_INITIALIZE (COUNT_WD, VERSIONWORD, BREAKS, RMTF);
2800     VALUE VERSIONWORD;
2900     REAL COUNT_WD, VERSIONWORD; ARRAY BREAKS [0]; FILE RMTF;
3000     LIBRARY ICD_LIBRARY;
3100
3105 PROCEDURE ICD_INITIALIZE (COUNT_WD, VERSIONWORD, RMT_IN, RMT_OUT);
3110     VALUE COUNT_WD, VERSIONWORD;
3115     REAL COUNT_WD, VERSIONWORD; FILE RMT_OUT, RMT_IN;
3120     LIBRARY ICD_LIBRARY;
3125
3200 PROCEDURE ICD_CALL (SEQNUM);
3250            VALUE SEQNUM;
3300            REAL SEQNUM;
3400     LIBRARY ICD_LIBRARY;
3500     % ************ End of ICD stuff ************
3600 PROCEDURE LEVEL_3_Procedure; EXTERNAL;
3700  EXPORT LEVEL_3_Procedure;
3800
3900  INTEGER I;
4010   WRITE (REMOTE_FILE
4020            , <"About to call ICDLIBVSN16 ALGOL_INIT">);
4030      % ************ Beginning of ICD stuff ************
4200   ICD_ALGOL_INITIALIZE(
4300        COUNT_WORD,    0,       BREAK_ARAY, REMOTE_FILE );
4301     % ************ End of ICD stuff ************
4302   % Set a global that we can try to access from bound rtns
4303 VERSION_WORD := 16.6;
4304
4500  I := 2;
4600  REMOTE_FILE.OPEN := TRUE;
4700  WRITE (REMOTE_FILE, <"Host opened Remote file">);
4900
5000  IF BOOLEAN(MYSELF.LIBRARYSTATE) THEN
5100         % just export the Level 3 Procedure and Freeze
5200     BEGIN %(*We're a LIB*)
5300     WRITE (REMOTE_FILE, <"Freezing as a Library">);
5400     FREEZE (TEMPORARY);
5500     END   %(*We're a LIB*)
5600  ELSE
5900     BEGIN %(*Normal invocation*)
6000     DISPLAY ("Called normally");
6100     LEVEL_3_Procedure;
6200     END;  %(*Normal invocation*)
6300 END.
[bookmark: _Toc34646225]Level 3 Subroutine
DEBUGGER/BINDING/POSSIBLELIB/LVL3

00000100$SET LEVEL 3
00000200[
00000300 FILE REMOTE_FILE;
00000400]
00000500
00000600PROCEDURE LEVEL_3_Procedure;
00000700  BEGIN %(*LEVEL_3_Procedure*)
00001200
00001300  PROCEDURE LEVEL_4_Procedure; EXTERNAL;
00001305
00001310 INTEGER I;
00001315
00001400
00001800   DISPLAY ("Calling Level 4 Procedure, Hit RETURN when ready");
00001850   READ (REMOTE_FILE,*,I);
00001900  LEVEL_4_Procedure;
00002000
00002100  END;  %(*LEVEL_3_Procedure*)
[bookmark: _Toc34646226]Level 4 subprogram
Note that when the host is run as a library the globals defined here are in the library stack at run-time. They are not in the caller's stack. ICD handles this.
DEBUGGER/BINDING/POSSIBLELIB/LVL4_COBOL74

100 $  SET ICD
200 $SET LEVEL = 4
300 IDENTIFICATION DIVISION.
400 PROGRAM-ID. ARRAY-MIXER.
500 ENVIRONMENT DIVISION.
600 INPUT-OUTPUT SECTION.
700 FILE-CONTROL.
800     SELECT GLOBAL REMOTE-FILE           ASSIGN TO REMOTE.
900 DATA DIVISION.
1000 FILE SECTION.
1100 FD REMOTE-FILE.
1200     01 REMOTE-FILE-REC      PIC X(80).
1325*
1400 WORKING-STORAGE SECTION.
1500*    A 77 REAL will get passed as REAL by REF
1600 77 LEVEL-NUMBER             REAL.
1700 77 VERSION-WORD   GLOBAL REAL.
1800 77 LEVEL-4-OWN-WORD OWN  REAL.
1900 PROCEDURE DIVISION.
2000 THAT SECTION.
2100 MIX.
2200    MOVE 4 TO LEVEL-NUMBER.
2300    MOVE 41 TO LEVEL-4-OWN-WORD.
2350*
5900    MOVE "In Cobol Level4 Procedur, Hit RETURN" TO
6000                 REMOTE-FILE-REC.
6100    WRITE REMOTE-FILE-REC.
6200    READ REMOTE-FILE.
6300 END-IT-ALL.
[bookmark: _Toc34646227]ALGOL bind cards
The bind deck when using an ALGOL host requires two extra USE cards to link the procedures ICD_CALL and ICD_INITIALIZE in the COBOL74 bound routine to those declared in the host. The statements are:
USE ICD_CALL FOR .ICD_CALL;
USE ICD_INITIALIZE FOR .ICD_INITIALIZE;
In this example only the LEVEL_4_PROCEDURE has ICD set so it is the only one that needs the USE cards.

HOST IS OBJECT/DEBUGGER/BINDING/POSSIBLELIB/HOST;
BIND LEVEL_3_PROCEDURE FROM
	OBJECT/DEBUGGER/BINDING/POSSIBLELIB/LVL3;
BIND LEVEL_4_PROCEDURE FROM
	OBJECT/DEBUGGER/BINDING/POSSIBLELIB/LVL4_COBOL74;
	USE ICD_CALL FOR .ICD_CALL;
	USE ICD_INITIALIZE FOR .ICD_INITIALIZE;
	USE REMOTE_FILE FOR REMOTE-FILE;
	USE VERSION_WORD FOR VERSION-WORD;

[bookmark: _Toc34646228]Binder Output
The BINDER output will be like:
 HOST IS OBJECT/DEBUGGER/BINDING/POSSIBLELIB/HOST ON SYMBOL - , DCALGOL 40.140,
BEGIN BINDING LEVEL_3_PROCEDURE OF BLOCK#1 FROM OBJECT/DEBUGGER/BINDING....
      LEVEL_3_PROCEDURE  (02,0003)  HAS BEEN CHANGED TO  (02,000C)
      REMOTE_FILE  (02,0002)  HAS BEEN CHANGED TO  (02,0006)
BEGIN BINDING LEVEL_4_PROCEDURE OF LEVEL_3_PROCEDURE OF BLOCK#1 FROM OBJECT/DEBUGGER/BINDING/POSSIBLELIB/"LVL4_COBOL74",
      COBOL74 42.152, COMPILED 09/04/95 13:03:56
THE IDENTIFIER OF THE SEPARATE PROCEDURE DOES NOT MATCH THE DECLARATION IN THE HOST -- THIS IS A WARNING ONLY <<<<
STATEMENT REQUIRED: USE LEVEL_4_PROCEDURE FOR CODE2170;
      REMOTE_FILE  (02,0002)  HAS BEEN CHANGED TO  (02,0006)
REMOTE-FILE/REMOTE-FILE-REC  <---- THIS NEW GLOBAL VARIABLE HAS BEEN ADDED TO THE HOST -- THIS IS A WARNING ONLY
      REMOTE-FILE/REMOTE-FILE-REC  (02,0003)  HAS BEEN CHANGED TO  (02,0011)
      VERSIONWORD  (02,0004)  HAS BEEN CHANGED TO  (02,0002)
      ICD_LIBRARY  (02,0006)  HAS BEEN CHANGED TO  (02,0007)
ICD_LIBRARY  <LIBRARY> THE LIBRARY ATTRIBUTES IN THE SUBPROGRAM DIFFER ...<<<<<
                       THE HOST LIBRARY ATTRIBUTES WILL BE USED. <<<<<<<<<<<<
      ICD_CALL  (02,0007)  HAS BEEN CHANGED TO  (02,000B)
      ICD_INITIALIZE  (02,0008)  HAS BEEN CHANGED TO  (02,000A)
      LEVEL-4-OWN-WORD  (02,0005)  HAS BEEN CHANGED TO  (02,0012)
END OF BINDING LEVEL_4_PROCEDURE
END OF BINDING LEVEL_3_PROCEDURE

NOTE: Ignore the message:
[bookmark: _GoBack]THE LIBRARY ATTRIBUTES IN THE SUBPROGRAM DIFFER FROM THE HOST.

